

Stata syntax for the estimation of the internal consistency of essentially unidimensional measures according to two compatible measurement models as described in

Viladrich, C., Angulo-Brunet, A., & Doval, E. (2017). A journey around alpha and omega to estimate internal consistency reliability. *Annals of Psychology*, 33(3), 755-782. doi: 10.6018/analesps.33.3.268401

See <http://ddd.uab.cat/record/173917> for the dataset headed with the variable names and Table 1 and Table 2 in the paper for selected output obtained running the first measurement model with R.

See <http://ddd.uab.cat/record/205870> for the dataset without variable names, as used in the present syntax.

Output using Stata could differ from Table 1 and Table 2 due to differences in computational algorithms.

Table of contents

*Starting Stata

*Case3: Essentially unidimensional measures

*Confirmatory bifactor measurement model

*Starting Stata

*Defining and checking the working directory

```
cd "C:\workingdirectory"
```

```
pwd
```

*Installing and checking packages needed to perform the analyses

*Don't run if already installed!

```
ado
```

```
ssc install tab_chi
```

```
net install polychoric, from(http://staskolenikov.net/stata)
```

```
ssc install relicoef
```

```
ado
```

*Case 3: Essentially unidimensional measures

*Reading data

```
infile y1 y2 y3 y4 y5 y6 using "Case3_noNames.txt", clear
```

*Phase 1

*Response percentages

```
tabm y1-y6, row
```

*Other univariate statistics

```
tabstat y1-y6, statistics( count min max mean sd skewness kurtosis ) noseparator columns(statistics)
```

*Pearson correlations

```
correlate y1-y6
```

*Measures with correlated errors

*Phase 2

```
sem (F1err_cor -> y1-y6), var(F1err_cor@1) cov(e.y4*e.y5) cov(e.y4*e.y6) cov(e.y5*e.y6) nolog
```

```
estat gof, stats(all)
```

```
sem,standardized
```

*Phase 3

*Alpha coefficient not included

*Point estimation of omega coefficient for measures with correlated errors

```
sem (F1err_cor -> y1-y6), var(F1err_cor@1) cov(e.y4*e.y5) cov(e.y4*e.y6) cov(e.y5*e.y6) nolog
```

```
relicoef
```

*Interval estimation not included

*Confirmatory bifactor measurement model

*Phase 2

*Specification, estimation and fit of the bifactor model

sem (Fg -> y1-y6) (Fs-> y4 y5 y6), var(Fg@1) var(Fs@1) cov(Fg*Fs@0)

estat gof, stats(all)

sem,standardized

*Phase 3

*point estimation of coefficient omega hierarchical and omega total for the general factor

sem (Fg -> y1-y6) (Fs-> y4 y5 y6), var(Fg@1) var(Fs@1) cov(Fg*Fs@0)

estat framework, compact

mat L=r(Gamma)

mat E=r(Psi)

scalar resid = trace(E)

mata: sla= colsum(st_matrix("L"))

mata: st_numscalar("gen", sla[1]^2)

mata:st_numscalar("spec", sla[2]^2)

scalar omegaHg=gen/(gen+spec+resid)

display "McDonald's omega(h) = " omegaHg

scalar omegaTg=(gen+spec)/(gen+spec+resid)

display "McDonald's omega(T) = " omegaTg

*Interval estimation not included